

Добавление произвольных реквизитов с контактной информацией

← → ☆ Демо: Заказ покупателя 00-00000004 от 04.06.201... 🔗 ⋮ ×

Основное [Взаимодействия](#) [Задачи](#) [Связанные документы](#)

Провести и закрыть 📄 📄 👤 Отправить 📄 📄 📄 ⋮ ?

Номер: 00-00000004 от: 04.06.2012 12:57:12 📅 Статус заказа: Согласован ▾

Партнер: ООО "Альфа" 📄 Организация: Новые технологии ООО 📄

Контрагент: ООО "Альфа" 📄 Договор: Договор Альфа с Новые технологи 📄

Сумма документа: 60 000,00 📄 Валюта: RUB ▾

Счета на оплату (1) **Контактные лица** Доставка

Доставить: . . . : : 📅

Контактное лицо: Либерзон Мария Робертовна ▾ 📄

Адрес: ...

Комментарий:

Электронная почта: 📧

Уведомлять об изменении статуса заказа Изменение статуса заказа ▾

Имеется возможность хранить значения полей контактной информации не только в стандартной табличной части КонтактнаяИнформация, но и в произвольных реквизитах произвольных объектов конфигурации. Такая возможность может быть востребована для добавления к объектам отдельных полей с контактной информацией.

Например, необходимо хранить данные адреса доставки в документе ЗаказПокупателя. При этом в форме документа должно быть поле **Адрес доставки**, редактируемое с помощью стандартной формы ввода адреса, и поля **Дата доставки**, **Метро**, **Комментарий**. При этом в заказе покупателя не требуются остальные сервисы подсистемы по заведению других видов контактной информации.

Для решения этой задачи необходимо:

1. Добавить к документу ЗаказПокупателя основные реквизиты:
 - АдресДоставки типа Строка неограниченной длины. Этот реквизит будет хранить служебные значения контактной информации адреса доставки;
 - ДатаДоставки типа Дата для хранения даты доставки;
 - Метро. Ссылка на справочник СтанцииМетро.
2. Добавить к документу ЗаказПокупателя вспомогательные реквизиты, данные в которых будут вычисляться при записи документа по данным реквизита АдресДоставки. Эти реквизиты предназначены для оперативного поиска документов по данным контактной информации, вывода данных на печать и т. п.
 - АдресДоставкиСтрокой – строка длиной 200 символов;

- ГородАдресаДоставки – строка длиной 100 символов.

3. Поместить в форму вспомогательный реквизит

ВидКонтактнойИнформацииАдресаДоставки для контроля работы с адресом и инициализировать его при создании формы:

```
1 ВидКонтактнойИнформацииАдресаДоставки = Новый Структура;
2 ВидКонтактнойИнформацииАдресаДоставки.Вставить("Тип", Перечисления.ТипыКонтактнойИнформации.Адрес);
3 ВидКонтактнойИнформацииАдресаДоставки.Вставить("ТолькоНациональныйАдрес", Ложь);
4 ВидКонтактнойИнформацииАдресаДоставки.Вставить("ВключатьСтрануВПредставление", Ложь);
5 ВидКонтактнойИнформацииАдресаДоставки.Вставить("СкрыватьНеактуальныеАдреса", Ложь);
```

<https://paste1c.ru/2jl2se7bol8r>

Эта структура заменяет собой соответствующий элемент справочника

ВидыКонтактнойИнформации.

4. Поместить в форму документа вспомогательный строковый реквизит

ПредставлениеАдресаДоставки. Инициализировать значение реквизита при создании формы:

```
ПредставлениеАдресаДоставки = УправлениеКонтактнойИнформацией.ПредставлениеКонтактнойИнформации(Объект.АдресДоставки);
```

<https://paste1c.ru/3phi2mn3cpzp>

- создать соответствующее вспомогательному реквизиту поле ввода, установить у него флажок КнопкаВыбора В Истина;
- определить обработчики событий для этого элемента:

&НаКлиенте

Процедура ПредставлениеАдресаДоставкиПриИзменении(Элемент)

```
Текст = Элемент.ТекстРедактирования;
Если ПустаяСтрока(Текст) Тогда
    ПредставлениеАдресаДоставки = "";
    КомментарийАдресаДоставки = "";
    Объект.АдресДоставки = "";
Возврат;
КонецЕсли;
ПредставлениеАдресаДоставки = Текст;
Объект.АдресДоставки = ЗначенияПолейКонтактнойИнформацииСервер(Текст,
    ВидКонтактнойИнформацииАдресаДоставки, КомментарийАдресаДоставки);
```

КонецПроцедуры

&НаКлиенте

Процедура ПредставлениеАдресаДоставкиНачалоВыбора(Элемент, ДанныеВыбора, СтандартнаяОбработка)

```
Если Элемент.ТекстРедактирования<>ПредставлениеАдресаДоставки Тогда
    ПредставлениеАдресаДоставки = Элемент.ТекстРедактирования;
    Объект.АдресДоставки = "";
КонецЕсли;

ПараметрыОткрытия = Новый Структура;
ПараметрыОткрытия.Вставить("ВидКонтактнойИнформации", ВидКонтактнойИнформацииАдресаДоставки);
ПараметрыОткрытия.Вставить("Значение", Объект.АдресДоставки);
ПараметрыОткрытия.Вставить("Представление", ПредставлениеАдресаДоставки);
ПараметрыОткрытия.Вставить("Комментарий", КомментарийАдресаДоставки);
ПараметрыОткрытия.Вставить("Заголовок", НСтр("ru=Адрес доставки"));

УправлениеКонтактнойИнформациейКлиент.ОткрытьФормуКонтактнойИнформации(ПараметрыОткрытия, Элемент);
```

КонецПроцедуры

```
&НаКлиенте
```

```
Процедура ПредставлениеАдресаДоставкиОчистка(Элемент, СтандартнаяОбработка)
```

```
ПредставлениеАдресаДоставки = "";  
КомментарийАдресаДоставки = "";  
Объект.АдресДоставки = "";
```

```
КонецПроцедуры
```

```
&НаКлиенте
```

```
Процедура ПредставлениеАдресаДоставкиОбработкаВыбора(Элемент, ВыбранноеЗначение, СтандартнаяОбработка)
```

```
СтандартнаяОбработка = Ложь;  
Если ТипЗнч(ВыбранноеЗначение) <> Тип("Структура") Тогда  
    // Отказ от выбора, данные неизменны  
    Возврат;  
КонецЕсли;  
  
ПредставлениеАдресаДоставки = ВыбранноеЗначение.Представление;  
КомментарийАдресаДоставки = ВыбранноеЗначение.Комментарий;  
Объект.АдресДоставки = ВыбранноеЗначение.КонтактнаяИнформация;
```

```
КонецПроцедуры
```

<https://paste1c.ru/348dgv1caega>

Следует заметить, что если функция ввода адреса строкой не требуется, то можно обойтись без этого вспомогательного поля ввода, заменив его, например, на гиперссылку. Текст гиперссылки при этом будет представлением адреса, а обработчик нажатия аналогичен обработчику НачалоВыбора.

5. Поместить в форму вспомогательный реквизит КомментарийАдресаДоставки для редактирования комментария, не попадающего в основное представление адреса. Инициализировать его при создании формы:

```
КомментарийАдресаДоставки =  
УправлениеКонтактнойИнформацией.КомментарийКонтактнойИнформации(Объект.АдресДоставки);
```

<https://paste1c.ru/92bail2oceg>

6. Поместить в форму поле ввода и связать его с реквизитом КомментарийАдресаДоставки. Определить следующие обработчики изменений:

```
&НаКлиенте
```

```
Процедура КомментарийАдресаДоставкиПриИзменении(Элемент)
```

```
    ЗаполнитьКомментарийАдресаДоставкиСервер();
```

```
КонецПроцедуры
```

```
&НаСервере
```

```
Процедура ЗаполнитьКомментарийАдресаДоставкиСервер()
```

```
    Если ПустаяСтрока(Объект.АдресДоставки) Тогда  
        Объект.АдресДоставки = ЗначенияПолейКонтактнойИнформацииСервер(ПредставлениеАдресаДоставки,  
            ВидКонтактнойИнформацииАдресаДоставки, КомментарийАдресаДоставки);  
        Возврат;
```

```
    КонецЕсли;
```

```
    УправлениеКонтактнойИнформацией.КомментарийКонтактнойИнформации(Объект.АдресДоставки, КомментарийАдресаДоставки);
```

```
КонецПроцедуры
```

<https://paste1c.ru/3580rz7dmunu>

7. Поместить в форму документа поля ввода, связанные с реквизитами объекта ДатаДоставки и Метро.

8. В модуле объекта документа, в обработчике события `ПередЗаписью` заполнить вспомогательные реквизиты:

`ГородДоставки = УправлениеКонтактнойИнформацией.ГородАдресаКонтактнойИнформации(АдресДоставки);`

`АдресДоставкиСтрокой = УправлениеКонтактнойИнформацией.ПредставлениеКонтактнойИнформации(АдресДоставки);`

<https://paste1c.ru/2fdynvj19u3u>

9. БСП обеспечивает программный интерфейс для оперирования данными:
- методы для разбора контактной информации из строки представления;
 - методы для формирования представления контактной информации;
 - методы для получения и установки значений частей контактной информации (например, комментария, страны адреса, города адреса и т. п.);
 - методы для сравнения объектов контактной информации.

Пример реализации см. в демонстрационной конфигурации в реквизите `АдресДоставки` документа `_ДемоЗаказПокупателя`.